REGULAR HEPTAGON'S MIDPOINTS CIRCLE

ZVONKO ČERIN

Abstract. This paper explores the geometry of the regular heptagon $ABCDEFG$. We start from a classical result by Thébault and Demir that six midpoints of sides and diagonals lie on a cirlce m with diameter equal to the side of the square inscribed in the circumcircle of $ABCDEFG$. Then we discover eight more midpoints of segments on m and show that they are vertices of two regular heptagons inscribed in the circle m. Extending further this idea we show that midpoints of many other segments also lie on the circle m so that it deserves the name – the *midpoints circle* of $ABCDEFG$. In the proofs we use the complex numbers and perform our calculations with the help of computers in Maple V.

The regular heptagon (i. e., the planar regular convex polygon with seven vertices) has not been studied extensively like its cousins the equilateral triangle, the square, the regular pentagon, and the regular hexagon. Perhaps the reason is because this is the first regular polygon that cannot be constructed only with compass and straightedge. The few sporadic known results on regular heptagons were reviewed 30 years ago by Leon Bankoff and Jack Garfunkel in the reference [1]. One of the simplest is the following result by Victor Thébault and Hüseyin Demir [1, p. 10] which shows that the midpoints of several segments in the regular heptagon are related to the inscribed square.

Theorem 1. *Let $ABCDEFG$ be a regular heptagon inscribed in a circle k of radius R and center O. If P is the midpoint of the shorter arc BC and U and V are midpoints of the segments AB and OP then $|UV| = \frac{\sqrt{2}}{2}R$. In other words, $2|UV|$ is equal to the side of the square inscribed in the circle k. Moreover, the midpoints of the segments BG, AE, DG, CE, and CD share this property with the point U.*

Proof. Without loss of generality we can assume that $R = 1$ and that the vertices F, G, A, B, C, D, and E correspond to 7th roots 1, e, e^2, e^3, e^4, e^5, and e^6 of unity. Then the points O, P, U, and V are at

1991 *Mathematics Subject Classification.* Primary 51N20, 51M04, Secondary 14A25, 14Q05.

Key words and phrases. regular heptagon, midpoint, circle.



Figure 1: Selection of affixes of points in the proof of Theorem 1.

complex numbers 0, -1, $\frac{e^2+e^3}{2}$, and $-\frac{1}{2}$ (see Figure 1). Hence,

$$|UV|^2 = (U-V)\overline{(U-V)} = (U-V)(\overline{U}-\overline{V}) =$$
$$\left(\frac{e^2+e^3}{2}+\frac{1}{2}\right)\left(\frac{e^5+e^4}{2}+\frac{1}{2}\right) = \frac{2+1+e+e^2+\cdots+e^6}{4} = \frac{1}{2}$$

because $1+e+e^2+\cdots+e^6 = 0$. The proofs for the other five midpoints are similar. □

The above proof is indeed simple. It uses the well-known identification of points and complex numbers in the Gauss plane, the fact that $\frac{X+Y}{2}$ is the midpoint of the segment XY, that the square of the distance between points X and Y is the product $(X-Y)\overline{(X-Y)}$ of $X-Y$ and its conjugate $\overline{X-Y}$, that the conjugation satisfies $\overline{X-Y} = \overline{X}-\overline{Y}$, some algebraic simplification and the special property of the 7th root of unity at the end.

With so many computers around us and their profound influence on our lives one can wonder if we can discover and prove Theorem 1 with some help from computers.

Our figures are made in the software The Geometer's Sketchpad, the tool that can be used for approximate verification of statements and in the discovery of new theorems in geometry.

For example, to make Figure 1, we draw the points O and A, the circle k from them and mark O as a center (of a rotation). Then we rotate for $\frac{360}{7}$ degrees the point A six times in succession to get the vertices of $ABCDEFG$. For the point P we rotate F about O for 180 degrees. The points V, U, U', W, W', L and L' are midpoints of segments OP, AB, CD, BG, CE, AE and DG. We see that the circle m constructed from the points V (as the center) and U goes through these midpoints. This is only an indication that Theorem 1 is true and it is not its proof because The Geometer's Sketchpad has maximal precision of hundred thousandths that falls short of absolute correctness.

Looking back at the above proof of Theorem 1 we see that the key step was the algebraic simplification. The software like Derive, Mathematica and Maple V excel in this task so that they will easily do this part provided we make some preparation.

We will now make this in Maple V first for Theorem 1 and then build up the necessary functions that will be used to discover and prove all our other results.

We give points as ordered pairs $[p, q]$ of the complex number p and its conjugate q. The complex number e is the 7th root of unity, i. e., $e^7 - 1 = 0$. From $e^7 - 1 = (e-1)(e^6 + e^5 + e^4 + e^3 + e^2 + e + 1)$ and $e \neq 1$ we get $\sigma = e^6 + e^5 + e^4 + e^3 + e^2 + e + 1 = 0$. Hence, we input:

```
hF:=[1,1]:  hG:=[e,e^6]:  hA:=[e^2,e^5]:  hB:=[e^3,e^4]:
hC:=[e^4,e^3]:  hD:=[e^5,e^2]:  hE:=[e^6,e]:  hP:=[-1,-1]:
hV:=[-1/2,-1/2]:
```

Here we use `hA` instead of `A` as a name of the first vertex because with plain letters we run into problems as some letters are reserved in Maple V (for example `D`).

In order to cut down typing we introduce the shortening `FS` for the simultaneous use of commands `factor` and `simplify` that will be used frequently.

```
FS:=x->factor(simplify(x)):
```

The following procedures `dis` and `mid` compute the square of the distance between two points a and x and their midpoint. The letters b, c, y, and z denote local variables. They are the first and the second coordinates of the given points. The square of the distance is the product of the difference $b - y$ and its conjugate $c - z$. The midpoint is $\frac{b+y}{2}$

and its conjugate is $\frac{c+z}{2}$.

```
dis:=proc(a,x) local b,c,y,z; b:=a[1]:  c:=a[2]:
     y:=x[1]:  z:=x[2]:  FS((b-y)*(c-z)):  end:

mid:=proc(a,x) local b,c,y,z; b:=a[1]:  c:=a[2]:
     y:=x[1]:  z:=x[2]:  FS([b/2+y/2, c/2+z/2]):  end:
```

The circle m is the locus of all points whose square of distance to the point V is equal to $\frac{1}{2}$. The following function `hm` associates to a point the difference of the square of its distance from V and $\frac{1}{2}$. A point T will lie on the circle m if and only if the value $hm(T)$ is zero.

```
hm:=x->FS(dis(x,hV)-1/2):
```

The proof of Theorem 1 amounts to check whether the values of the function hm in the points U, U', W, W', L and L' are zero.

```
hU:=mid(hA,fB): hUp:=mid(hC,fD): hW:=mid(hB,fG):
hWp:=mid(hC,fE): hL:=mid(hA,fE): hLp:=mid(hD,fG):
hm(hU); hm(hUp); hm(hW); hm(hWp); hm(hL); hm(hLp);
```

The outputs are $\frac{\alpha\,\sigma}{4}$, $\frac{\alpha\,\sigma}{4}$, $\frac{\beta\,\sigma}{4}$, $\frac{\beta\,\sigma}{4}$, $\frac{\gamma\,\sigma}{4}$ and $\frac{\gamma\,\sigma}{4}$, where $\alpha = e^2 + e - 1$, $\beta = e^3 - e^2 + 2e - 1$ and $\gamma = e^5 - e^4 + 2e - 1$. Since they all contain σ as a factor and $\sigma = 0$ we conclude that these midpoints are on the circle m and the proof of Theorem 1 in Maple V is accomplished.

In this note we shall first add six new segments whose midpoints also lie on the circle m with the center at the point V and the radius $\frac{R\sqrt{2}}{2}$. The discovery of these new points was by chance while playing with the Sketchpad. However, the symmetry of Figure 1 in the line FO and the fact that the intersections H, I, J of the lines AE, AB, BG with their reflections DG, CD, CE are on FO make these intersections obvious candidates for endpoints of such segments.

Theorem 2. *Let $ABCDEFG$ be a regular heptagon inscribed in a circle k of radius R and center O. Let $H = AE \cap DG$, $I = AB \cap CD$, and $J = BG \cap CE$. If P is the midpoint of the shorter arc BC and V is the midpoint of the segment OP then the midpoints X, Y, U, W, K, L, L', K', W', U', Y', and X' of the segments BI, GJ, AB, BG, AH, AE, DG, DH, CE, CD, CI, and EJ, respectively, all lie on the circle m. The line joining the intersections Q and Q' of the circles k and m is the perpendicular bisector of the segment PV.*

Proof. Let the assumptions of the proof of Theorem 1 hold. Then the equation of the circle m is $\left[z - (-\frac{1}{2})\right]\left[\overline{z} - (-\frac{1}{2})\right] = \frac{1}{2}$ which simplifies



Figure 2: The circle m is passing through 14 midpoints.

to $4\,z\,\bar{z} + 2\,z + 2\,\bar{z} - 1 = 0$. The method of proof is to find the complex coordinate of each point and check that they satisfy this equation. For example, in order to do this for the point X (the midpoint of the segment joining the vertex B with the intersection I of lines AB and CD), we determine equations of lines AB and CD and solve them to get $I = \frac{(e+1)e^4}{e^2+1}$ and $X = \frac{e^3\left(2\,e^2+e+1\right)}{2(e^2+1)}$. Let δ denote $2\,e^5 + e^4 + 3\,e^3 - 2\,e^2 + e - 1$. Since $4\,X\,\bar{X} + 2\,X + 2\,\bar{X} - 1$ factors as $\frac{\delta\,\sigma}{(e^2+1)^2}$, we conclude that the point X is on the circle m.

The last claim is true because the circles m and k (whose equation is $z\,\bar{z} = 1$) intersect in points $Q = -\frac{3}{4} + i\,\frac{\sqrt{7}}{4}$ and $Q' = -\frac{3}{4} - i\,\frac{\sqrt{7}}{4}$. □

In order to implement the above proof in Maple V we need the functions for lines and their intersections.

We represent lines as ordered triples $[u,\ v,\ w]$ of coefficients of their equations $u\,z + v\,\bar{z} + w = 0$. The procedure `lin` gives the line through two different points.

```
lin:=proc(m,n) local a,b,c,x,y,z; a:=m[1]:  b:=m[2]:
       x:=n[1]:  y:=n[2]:  FS([b-y, x-a, a*y-b*x]):  end:
```

That the coefficients u, v, and w are indeed $b - y$, $x - a$, and $ay - bx$ follows easily by substituting the coordinates of the points m and n

and solving in u and v.

```
solve({u*a+v*b+w,u*x+v*y+w},{u,v});
```

The output is $\{u = \frac{w\,(b-y)}{a\,y-b\,x},\ v = \frac{w\,(x-a)}{a\,y-x\,b}\}$. If we substitute this into the equation of the line and multiply with $ay - xb$ and divide with w we get the above form.

In a similar way we derive the important procedure which gives the intersection of two lines. When its usage results in the error message

```
Error, numeric exception:  division by zero
```

this means that $ay - xb = 0$ so that the lines are parallel (when they do not have an intersection).

```
inter:=proc(m,n) local a,b,c,x,y,z; a:=m[1]:
       b:=m[2]:  c:=m[3]:  x:=n[1]:  y:=n[2]:  z:=n[3]:
       FS([(b*z-c*y)/(a*y-b*x),(c*x-a*z)/(a*y-b*x)]):  end:
```

The verification of Theorem 2 is done with the following input:

```
fp:=(a,b,c,d)->inter(lin(a,b),lin(c,d)):
hH:=fp(hA,hE,hD,hG): hI:=fp(hA,hB,hC,hD): hX:=mid(hI,hB):
hJ:=fp(hB,hG,hC,hE)): hY:=mid(hJ,hG): hK:=mid(hH,hA):
hKp:=mid(hH,hD): hYp:=mid(hJ,hE): hXp:=mid(hI,hC):
hm(hX);hm(hY);hm(hK);hm(hKp);hm(hYp);hm(hXp);
```

The first output is $\frac{\delta\,\sigma}{(e^2+1)^2}$. This completes the proof for the point X.

Next we prove that the intersections Q and Q' of the circles k and m are also midpoints of segments. This observation is a consequence of our desire to have points Q and Q' as midpoints of some segments related to the regular heptagon as other twelve points from Theorem 2 are. The search for the points M and M' in the Sketchpad begins with any point S and its reflection S' in the point Q. We move S into various positions (for example, into the point H) and look carefully on what lines the point S' will be. Of course, a bit of luck is needed here but if you are patient the reward will come.

Theorem 3. *Let M and M' be the intersections of the lines AG and DE with the tangents to the circle k at the points C and B. Then Q and Q' are the midpoints of the segments HM and HM' (see Figure 2).*

Proof. The line AG has the equation $e^5\,z + e\,\overline{z} - e^6 - 1 = 0$ and the equation of the tangent to the circle k at C is $e^3\,z + e^4\,\overline{z} - 2 = 0$. They intersect in the point $M = \frac{e^4(e^2+2\,e+2)}{e^4+e^3+e^2+e+1}$. Now we can show that the

midpoint of the segment HM is the complex number

$$\frac{5\,e^6+5\,e^5+3\,e^4+2\,e^3+2\,e^2+e+2}{4(e^4+e^3+e^2+e+1)(e+1)}$$

and agrees with the point Q by finding that their distance is zero or by checking that it satisfies the equations of k and m. □

In Maple V the above proof begins with the procedure for the perpendicular through a given point to a given line:

```
per:=(t,p)->FS([p[1],-p[2],t[2]*p[2]-t[1]*p[1]]):
```

The following input defines the point M and verifies if the midpoint of HM is Q.

```
hM:=inter(lin(hA,hG),per(hC,lin(hC,hO))):
sQ:=solve({4*p*q+2*p+2*q-1, p*q-1}, {p,q}):
hQ:=subs(sQ[1],[p,q]):  hQp:=subs(sQ[2],[p,q]):
dis(hQ,mid(hH,hM));
```

The output is $\frac{\lambda\,\mu\,\nu^2}{64(e^4+e^3+e^2+e+1)^2(e+1)^2}$, where λ and μ are polynomials of order 7 in e and $\nu = 2e^3+e^2-e-2-e(e+1)k$ with $k=i\sqrt{7}$. But, the expression $\left[\frac{2e^3+e^2-e-2}{e(e+1)}\right]^2+7$ factors as $\frac{4\,\sigma}{e^2(e+1)^2}$. This implies that $2e^3+e^2-e-2=\pm e(e+1)k$. The minus sign is eliminated be looking at close enough numerical values. Hence, $\nu=0$ and Q is the midpoint of HM. More rigorous is the following direct proof that $\nu=0$.

```
hn:=2e^3+e^2-e-2-e(e+1)*I*sqrt(7):  t:=2*Pi/7:
FS(numer(convert(FS(subs(e=cos(t)+I*sin(t),hn)),exp)));
```

In the statement of Theorems 2 and 3 we described fourteen points on the circle m. These points are rather special because in the next result we will show that they are vertices of two regular heptagons.

In the proof of the first part of Theorem 4 we will use the following lemma.

Lemma 1. *Let* $\vartheta=\frac{\pi}{14}$. *Then* $8\cos^3\vartheta-4\sqrt{7}\cos^2\vartheta+\sqrt{7}=0$.

Proof. Let $u=\sin\vartheta$ and $v=\cos\vartheta$. From $\sin x=\sin(\pi-x)$ for $x=5\vartheta$ we get $\sin 5\vartheta=\sin 9\vartheta$. The left hand side is $\sin 5\vartheta=u(16\,v^4-12\,v^2+1)$ while the right hand side is

$$\sin 9\vartheta=u(256\,v^8-448\,v^6+240\,v^4-40\,v^2+1).$$

Now divide both sides by u and move all terms on one side to get $4\,v^2\,(64\,v^6-112\,v^4+56\,v^2-7)=0$. It follows that the polynomial in

the parenthesis is equal to zero. However, it is the product

$$64\,v^6 - 112\,v^4 + 56\,v^2 - 7 = (8\,v^3 - 4\sqrt{7}\,v^2 + \sqrt{7})(8\,v^3 + 4\sqrt{7}\,v^2 - \sqrt{7}).$$

Since the second parenthesis is not zero (because $v \in [\frac{3}{4}, 1]$), the first will be zero. □

The two lines of input for the proof of Lemma 1 in Maple V are:

```
t:=cos(Pi/14):  s:=sqrt(7):
FS(numer(convert(8*t^3-4*s*t^2+s,exp))):
```

Theorem 4. *(a) The polygons $XQWLK'U'Y'$ and $YUKL'W'Q'X'$ are regular heptagons inscribed in the circle m (see Figure 3).*

(b) If X'', Q'', W'', L'', K'', U'', and Y'' denote midpoints of the shorter arcs XY, QU, WK, LL', $K'W'$, $U'Q'$, and $Y'X'$ then the regular heptagon $Q''W''L''K''U''Y''X''$ has sides parallel to the corresponding sides of the heptagon $ABCDEFG$ (see Figure 4).



Figure 3: Two regular heptagons of midpoints inscribed in m.

Proof of Theorem 4(a). Since the vertices of heptagons $XQWLK'U'Y'$ and $YUKL'W'Q'X'$ are on the circle m, it suffices to show that all sides of these heptagons have the same length. When we compute the six differences of squares of distances $|YU|^2 - |UK|^2$, $|YU|^2 - |KL'|^2$, $|YU|^2 - |L'W'|^2$, $|YU|^2 - |W'Q'|^2$, $|YU|^2 - |Q'X'|^2$, $|YU|^2 - |X'Y|^2$ for the second heptagon, we shall always get zero. For the two differences involving Q' we do not get zero directly. In fact, we obtain expressions in sine and cosine functions which could be reduced so that they have $8\,(\cos\frac{\pi}{14})^3 - 4\sqrt{7}\,(\cos\frac{\pi}{14})^2 + \sqrt{7}$ as a factor. However, by Lemma 1,



Figure 4: The regular heptagon of midpoints of small arcs.

this factor is equal to zero too. The proof for the first heptagon is similar. □

Proof of Theorem 4(a) in Maple V. The six differences are computed with the following input:

```
fq:=(a,b)->FS(dis(hY,hU)-dis(a,b)):  fq(hU,hK);fq(hK,hLp);
fq(hLp,hWp); fq(hWp,hQp); fq(hQp,hXp); fq(hXp,hY);
```

The output of `op(3,fq(hWp,hQp));` is the expression ν that we already noted is equal to zero. □

Proof of Theorem 4(b). The composition of the dilation $D(O, \frac{\sqrt{2}}{2})$ with the translation by the vector $\overrightarrow{OV}$ transforms the heptagon $ABCDEFG$ into the unique heptagon inscribed in the circle m whose sides are parallel to the corresponding sides of $ABCDEFG$. Hence, it suffices to show that the midpoint of the shorter arc LL' is precisely the intersection $\frac{\sqrt{2}-1}{2}$ of m and the segment FO. This follows from the observation that $L = \frac{e^2(e^4+1)}{2}$ and $L' = \frac{e(e^4+1)}{2}$ are conjugates. □

In the rest of this note we wish to show that midpoints of many other segments also lie on the circle m.

In order to simplify our statements we use the following notation. The midpoint of points P_1 and P_2 is $[P_1,\ P_2]$ while $[P_1,\ \ell]$, $r(P_1,\ \ell)$, and $r(P_1,\ P_2)$ are the perpendicular to the line ℓ through the point P_1 and the reflections of P_1 in the line ℓ and in the point P_2, respectively.

The following are the results on other midpoints that lie on the circle m. The proofs of all of them are left to the reader because they are similar to the proof of Theorems 2 and 3, namely we identify the complex coordinate of the point and check that it satisfies the equation of m.

Theorem 5. *Let $T = [F,\ Q]$ and $T' = [F,\ Q']$. Let $S = [O,\ N]$ and $S' = [O,\ N']$, where $N = OT' \cap [V,\ OP]$ and $N' = OT \cap [V,\ OP]$. Then the points T, T', S and S' are on the circle m. Moreover, S and S' are antipodal to Q' and Q (see Figure 5).*



Figure 5: The midpoints T, T', S and S' lie on the circle m.

The following input into Maple V is checking Theorem 5:

```
hT:=mid(hF,hQ):hTp:=mid(hF,hQp):hm(hT);hm(hTp); ft:=
x->inter(lin(hO,x),per(hV,lin(hO,hP))): hN:=ft(hTp):
hNp:=ft(hT):hS:=mid(hO,hN):hSp:=mid(heO,hNp):hm(hS);
hm(hSp); dis(hV,mid(hQ,hSp)); dis(hV,mid(hQp,hS));
```

Theorem 6. *Midpoints* $[A,\, r(O,\, BC)]$, $[H,\, r(O,\, AB)]$, $[I,\, r(O,\, FG)]$, $[C,\, r(O,\, EG)]$, $[B,\, r(O,\, DE)]$, $[I,\, r(O,\, CE)]$, *and* $[D,\, r(O,\, Y)]$ *are the vertices of the regular heptagon* $A_1A_2A_3A_4A_5A_6A_7$ *inscribed in the circle* m *(see Figure 6).*



Figure 6: The first regular heptagon of the midpoints inscribed in the circle m (Theorem 6).

The following Maple V procedures give the reflection of a point in a line and in a point, respectively.

```
ref:=proc(t,p) local x,y,a,b,c; x:=t[1]:y:=t[2]:a:=p[1]:
b:=p[2]:  c:=p[3]:  FS([-(b*y+c)/a, -(a*x+c)/b]):  end:
```

```
rep:=proc(t,p) local x,y,a,b; x:=t[1]:y:=t[2]:
a:=p[1]:  b:=p[2]:  FS([-x+2*a, -y+2*b]):  end:
```

The points $A_1, \ldots, A_7$ are:

```
fu:=(a,b,c,d)->mid(a,ref(b,lin(c,d))):hA1:=fu(hA,hO,hB,hC):
fv:=(a,b,c)->mid(a,rep(b,c)):  hA2:=fu(hH,hO,hA,hB):
hA3:=fu(hI,hO,hF,hG): hA4:=fu(hC,hO,hE,hG): hA5:=
fu(hB,hO,hD,hE): hA6:=fu(hI,hO,hC,hE): hA7:=fv(hD,hO,hY):
```

Now we check that these points are on m and that by rotating the vertex A_7 for $k\frac{2\pi}{7}$ radians for $k = 1, \ldots, 6$ we get points $A_1, \ldots, A_6$.

```
for i from 1 to 7 do hm(hA||i); od; for i from 1 to 6 do
```

```
FS(e^i*(hA7[1]-hV[1])+hV[1]-hA||i[1]); od;
```

Theorem 7. *Midpoints* $[D,\, r(O,\, BC)]$, $[H,\, r(O,\, CD)]$, $[I,\, r(O,\, EF)]$, $[B,\, r(O,\, EG)]$, $[C,\, r(O,\, AG)]$, $[I,\, r(O,\, BG)]$, *and* $[A,\, r(O,\, Y')]$ *are the vertices of the regular heptagon* $B_1B_2B_3B_4B_5B_6B_7$ *inscribed in the circle* m *(see Figure 7).*



Figure 7: The second regular heptagon also of the midpoints of the segments connecting the reflected points (Theorem 7).

This time in Maple V, the points $B_1, \ldots, B_7$ are:

```
hB1:=fu(hD,hO,hB,hC): hB2:=fu(hH,hO,hC,hD): hB3:=
fu(hI,hO,hE,hF): hB4:=fu(hB,hO,hE,hG): hB5:=
fu(hC,hO,hA,hG):hB6:=fu(hI,hO,hB,hG):hB7:=fv(hA,hO,hYp):
```

Now we check that these points are on m and that by rotating the vertex B_1 for $k\frac{2\pi}{7}$ radians for $k = 1, \ldots, 6$ we get points $B_2, \ldots, B_7$.

```
for i from 1 to 7 do hm(hB||i); od; for i from 2 to 7 do
FS(e^(i-1)*(hB1[1]-hV[1])+hV[1]-hB||i[1]); od;
```

Theorem 8. *Midpoints* C_1, C_2, C_3, C_4, C_5, C_6 *and* C_7 *of the shorter arcs* B_1A_6, B_2A_5, B_3A_4, B_4A_3, B_5A_2, B_6A_1 *and* B_7A_7 *of the circle* m *are the vertices of the regular heptagon. The sides of* $C_1C_2C_3C_4C_5C_6C_7$

are parallel with the corresponding sides of the heptagon $GABCDEF$. $A_1A_2A_3A_4A_5A_6A_7$, $B_1B_2B_3B_4B_5B_6B_7$ *and* $C_1C_2C_3C_4C_5C_6C_7$ *are the images of* $K'U'Y'XQWL$, $KUYX'Q'W'L'$ *and* $W''Q''X''Y''U''K''L''$ *under the homothety* $H(V, -1)$ *(see Figure 8).*



Figure 8: The third regular heptagon of the midpoints of the shorter arcs (Theorem 8).

The point C_7 is $\frac{-1-\sqrt{2}}{2}$. We check that it is on m and that it has the same distance from A_7 and B_7. The rest is a routine verification that we leave to the reader as an exercise.

References

[1] Leon Bankoff and Jack Garfunkel, *The heptagonal triangle*, Mathematics Magazine, **46** (1973), 163–187.

Kopernikova 7, 10010 Zagreb, CROATIA, Europe
E-mail address: cerin@math.hr